
CTTITEM_D.sps and CTTITEM_P.sps were developed for dichotomous item analysis 

and polytomous item analysis respectively. Use CTTITEM_D.sps for dichotomously scored 
items, and CTTITEM_P.sps for polytomously scored items. The following instructions on 
how to use the syntax apply to both cases.   

 
How to run CTTITEM_D and CTTITEM_P 

 
Input data 

Input data should be in text format. The first line should contain the correct options for 
dichotomous items and the number of score categories for polytomous items (If you are 
analyzing data of the Likert type and need reverse scoring, the program assumes that your 
responses/scores for the polytomous items start from 1 with an interval of 1). For 
dichotomous items, item response data start from the second line with each line representing 
one case. For polytomous items, the second line instructs the program whether to reversely 
score the items (0) or not (1). Examinee scores start from the third line for polytomous items. 
Use a space for a missing value. Missing value will be treated as incorrect for the calculation 
of item and test statistics for dichotomous items. Listwise deletion is used for missing values 
in polytomous items.  

The program will score dichotomous items as right (coded as ‘1’) or wrong (coded as 
‘0’) based on the answer keys provided by the user. For polytomous items, the program will 
reversely score the items instructed by the user (i.e., ‘0’s on the second line of the data file). 
For example, if the user would like to have high scores to represent high levels of the trait 
being measured, then a ‘1’ should be entered on the second line of the data file for all 
positively stated items and a ‘0’ for all negatively stated items. See MC.txt and polyitem.dat 
for examples of input data.  
 
Syntax 
A few changes need to be made to the syntax to fit the current data structure. They have been 
highlighted in bold in the syntax files included in the Appendices and are briefly described 
below.  
1. File locations.  

There are several file paths in the syntax. The first one specifies the location of the data 
file; all other file paths specify where you want to store and retrieve the intermediate results. 
We recommend that you use the same folder for the data as well as the output files. The 
easiest way to change all the file paths is to replace all “c:\temp” with the file path for your 
designated analysis folder using the SPSS “replace all” function.  

2. Variable names:  
   In addition to the file paths, you need to change the variable names to fit the current 
data structure. They have all been highlighted, starting from the first one in the data list 
statement at the beginning of the syntax, 
 
DATA LIST FILE='C:\temp\MC.TXT' 

   / MC1 TO MC15 1-15. 

 
“MC1 to MC15” assign names to the variables to be read from “MC.TXT”, and “1-15” 
indicate the location of the variables in the dataset. The example used one column for each 
variable, starting from the 1st to the 15th column.  
Note: “MC1 TO MC15” can be used because all the variables are labeled consecutively. If the 
variables are not consecutively named, you should specify them separately (e.g., MC2 MC5 
MC10 MC6). 
 
  Use the same rule to modify other places that have been highlighted. Make sure that 
variable names should be consistent across the syntax. Notice that variable list (e.g., MC1 to 
MC15) does not work in the CROSSTABS statement. You should list all the variable names 
(e.g., MC1 MC2 MC3 etc.) for the CROSSTABS statement at the end of the CTTITEM_D 
syntax file. 
 

CROSSTABS 

  /TABLES=GRP BY MC1 MC2 MC3 MC4 MC5 MC6 MC7 MC8 MC9 MC10 MC11 MC12 MC13 

 MC14 MC15 

 

Note: this syntax file does not calculate biserial correlation. Enzmann (2002) has written a 
SPSS macro to compute biserial correlation. It is accessible through 
http://www2.jura.uni-hamburg.de/instkrim/kriminologie/Mitarbeiter/Enzmann/Software/Enz
mann_Software.html 
 
Output 

Outputs produced from our example data are available from the package. See 
MC_Output.spo for dichotomous items and polyitem_Output.spo for polytomous items. 

The first part of the output includes the statistics of the total score distribution. Note that 
a normal distribution has a kurtosis of 0 in SPSS.  

Descriptive Statistics

100 5.00 15.00 12.5900 2.43333 -1.054 .241 .571 .478
100

TOTAL SCORE
Valid N (listwise)

Statistic Statistic Statistic Statistic Statistic Statistic Std. Error Statistic Std. Error
N Minimum Maximum Mean Std. Skewness Kurtosis

 
 

The reliability analysis results that follow include Cronbach’s alpha, summary statistics 
for inter-item correlations, and item-total statistics. The “corrected item-total correlation” is 
commonly called the corrected point-biserial correlation.  

 Reliability Statistics 
 

Cronbach's 

Alpha 

Cronbach's Alpha Based 

on Standardized Items N of Items 

.727 .743 15

 

Summary Item Statistics

.161 -.123 .652 .775 -5.298 .020 15Inter-Item Correlations
Mean Minimum Maximum Range

Maximum /
Minimum Variance N of Items

The covariance matrix is calculated and used in the analysis.
 

 

Item-Total Statistics

11.6300 5.569 .337 . .716
11.6000 5.798 .235 . .725
11.6200 5.632 .319 . .719
11.6900 5.650 .125 . .731
11.7400 5.204 .359 . .710
11.8000 5.010 .406 . .704
11.7600 5.134 .377 . .708
11.6700 5.415 .340 . .713
11.8800 4.975 .363 . .710
11.8600 5.213 .250 . .725
11.9400 4.966 .339 . .714
11.7100 5.178 .428 . .703
11.8100 5.044 .376 . .708
11.7600 4.952 .492 . .694
11.7900 5.137 .341 . .712

MC1
MC2
MC3
MC4
MC5
MC6
MC7
MC8
MC9
MC10
MC11
MC12
MC13
MC14
MC15

Scale Mean if
Item Deleted

Scale
Variance if

Item Deleted

Corrected
Item-Total
Correlation

Squared
Multiple

Correlation

Cronbach's
Alpha if Item

Deleted

 
P value and D-index are then listed for every item in its original order. (For polytomous 

items, p value is replaced by item mean and D_INDEX does not apply).  
 

ITEM_NUM   PVALUE  D_INDEX 

      1       .96      .10 

      2       .99      .03 

      3       .97      .10 

      4       .90      .21 

      5       .85      .34 

      6       .79      .48 

      7       .83      .41 

      8       .92      .21 

      9       .71      .59 

     10       .73      .48 

     11       .65      .72 

     12       .88      .41 

     13       .78      .45 

     14       .83      .55 

     15       .80      .48 

 
If your data are dichotomously scored, frequencies of options endorsed by the low- and 

high- scoring groups (distractor analysis) will be displayed.  

  

GROUP * MC3 Crosstabulation

Count

29 0 0 29
26 1 2 29
55 1 2 58

HIGH
LOW

GROUP

Total

1 3 4
MC3

Total

 

 

Note that only the options that are endorsed by at least one examinee in any of the two groups 
will appear in this cross table.  
 

If your data are polytomously scored, score point frequencies for every item will be 
available. 

IT1 

  Frequency Percent Valid Percent 

Cumulative 

Percent 

1 7 8.8 8.8 8.8 

2 22 27.5 27.5 36.3 

3 21 26.3 26.3 62.5 

4 13 16.3 16.3 78.8 

5 8 10.0 10.0 88.8 

6 9 11.3 11.3 100.0 

Valid 

Total 80 100.0 100.0  

 
 In addition to the above statistics available from the output window, score information 

is also saved in separate datasets for later use. For dichotomous items, ITEMSCORE.sav and 
TOTSCORE.sav contain the item score and total score of every subject; PD.sav includes p 
value and D-index statistics for every item. For polytomous items, ITEMSCORE.sav contains 
information for both item score and total score.   

 
Interpreting item statistics 

Item difficulty. For dichotomous items, p-values range from 0 to 1 with a higher value 
indicating easier items. Mid range p-values (around .50) are desired for norm-referenced tests 
because they tend to produce larger score variances which in turn lead to higher score 
reliability estimates. For multiple-choice or true-false items where guessing may be a factor, 

the optimal p-value is around .50.50
k

+ , where k is the number of options and .5/k is added to 

adjust for random guessing (e.g., Crocker & Algina, 1986, p.313). Because examinees do not 
actually guess randomly but eliminate options first before guessing, the actual optimal 

p-value is slightly higher than .50.50
k

+ .  

Discrimination indices. The D-index ranges from -1 to 1 with a higher value indicating 
stronger discriminating power. According to Ebel and Frisbie (1991), an item that has a 
D-index of .4 or higher is satisfactory, between .30 and .39 requires little or no revision, 
between .20 and .29 needs revision, and lower than .19 should probably be eliminated or 
completely revised. Point-biserial and biserial provide the item-total correlation information, 
and their difference is that biserial assumes a normal distribution of ability underlying the 
dichotomous responses. Biserial correlation is constantly larger than its point biserial 
counterpart (by at least 25%) and can be larger than one when the total scores are not 
normally distributed (e.g., Cohen, Cohen, West, & Aiken, 2003, p. 57). Biserial or 
point-biserial correlations should not be negative if the items are discriminating in the proper 
direction. Furthermore, Emslie and Emslie (2002) suggested that an adequately 
discriminating item should have corrected point-biserial correlation of .3 or larger (which is 
roughly equivalent to biserial correlation of .4 or larger). 

Distractor analyses. For multiple-choice items, analysis of distractors can also provide 
useful information. If a multiple-choice item is functioning properly, more high-scoring than 
low-scoring examinees should choose the correct option, and the distractors should be 
equally plausible to examinees who do not know the correct answer (e.g., Aiken, 1997, 
p.65-85). Distractor analysis is often helpful in identifying miskeyed items (e.g., the 
high-scoring group selects a “distractor” more frequently than the “keyed” option), 
ambiguous options (e.g., the high-scoring group selects a distractor as frequently as the keyed 
option), blind guessing (e.g., responses of the high-scoring group distribute evenly among the 
alternatives), or examinee misconceptions (see Ebel & Frisbie, 1991 for more detailed 
discussion of option analysis). 

Appendix A 
CTTITEM_D.sps -- SPSS syntax file for dichotomous items 

DATA LIST FILE='C:\temp\MC.TXT' 

   / MC1 TO MC15 1-15. 

 

MISSING VALUES ALL ( ). 

SET MXLOOPS=300 /MPRINT=NO. 

*****************************************************************************. 

*Change variable names, be consistent with the previous statement. 

*****************************************************************************. 

MATRIX. 

GET RAW 

  /VARIABLES=MC1 TO MC15  

  /FILE=*  

  /MISSING=ACCEPT 

  /SYSMIS=-9. 

 

COMPUTE NCASE=NROW(RAW)-1. 

COMPUTE NITEM=NCOL(RAW). 

COMPUTE SCORE=MAKE(NCASE,NITEM,0). 

COMPUTE RAWN=MAKE(NCASE,NITEM,0). 

 

/*******SCORING ITEMS***********************/ 

LOOP N=2 TO NROW(RAW). 

- LOOP I=1 TO NITEM. 

-  DO IF(RAW(N,I)=RAW(1,I)). 

-  COMPUTE SCORE((N-1),I)=1. 

  ELSE. 

- COMPUTE SCORE((N-1),I)=0. 

- END IF. 

-END LOOP. 

END LOOP. 

 

LOOP N=2 TO NROW(RAW). 

- LOOP I=1 TO NITEM. 

-  COMPUTE RAWN((N-1),I)=RAW(N,I). 

-END LOOP. 

END LOOP. 

 

/******COMPUTE TOTAL SCORE********/ 

**************************************************************************. 

*Change file location('C:\temp') in the following several statements. 

*It is recommended that you put all datasets in the same directory. 

***************************************************************************. 

COMPUTE TOTSC=RSUM(SCORE). 

SAVE TOTSC 

/OUTFILE='C:\temp\TOTSCORE.SAV' 

/VARIABLES=TOTSC. 

 

*******************************************************************************************. 

*Change variable names 'MC1 TO MC15', be consistent with previous statements. 

*******************************************************************************************. 

SAVE SCORE 

/OUTFILE='C:\temp\ITEMSCORE.SAV' 

/VARIABLES=MC1 TO MC15. 

 

COMPUTE R=GRADE(TOTSC). 

COMPUTE RSC={R,TOTSC}. 

COMPUTE GRP=MAKE(NCASE,1,0). 

COMPUTE #LOW=0. 

COMPUTE LOWCUT=0. 

 

LOOP I=1 TO NCASE. 

-DO IF(RSC(I,1)/NCASE<=0.27) AND (LOWCUT<RSC(I,2)). 

  COMPUTE LOWCUT=RSC(I,2). 

  ELSE. 

-END IF. 

END LOOP. 

 

LOOP I=1 TO NCASE. 

-DO IF(RSC(I,2)<=LOWCUT). 

-COMPUTE #LOW=#LOW+1. 

-COMPUTE GRP(I,1)=3. 

 ELSE. 

-END IF. 

END LOOP. 

 

COMPUTE UPCUT=NITEM. 

 

LOOP I=1 TO NCASE. 

-DO IF(RSC(I,1)/NCASE>=0.73) AND (UPCUT>RSC(I,2)). 

  COMPUTE UPCUT=RSC(I,2). 

  ELSE. 

-END IF. 

END LOOP. 

COMPUTE #UP=0. 

 

LOOP I=1 TO NCASE. 

-DO IF(RSC(I,2)>=UPCUT). 

- COMPUTE GRP(I,1)=1. 

- COMPUTE #UP=#UP+1. 

  ELSE. 

-END IF. 

END LOOP. 

 

***************************************************************************. 

*Change file location 'C:\temp' and variable names 'MC1 to MC15'. 

*Be consistent with previous statements. 

**************************************************************************. 

COMPUTE RAWGRP={GRP,RAWN}. 

SAVE RAWGRP 

/OUTFILE='C:\temp\RAWGRP.SAV' 

/VARIABLES=GRP MC1 TO MC15. 

 

COMPUTE ITEMGRP={GRP,SCORE}.  

COMPUTE DINDEX=MAKE(NITEM,1,-9). 

COMPUTE TOHI=MAKE(NITEM,1,0). 

COMPUTE TOLO=MAKE(NITEM,1,0). 

 

LOOP I=1 TO NITEM. 

-LOOP N=1 TO NCASE. 

-DO IF(ITEMGRP(N,1)=1). 

-      COMPUTE TOHI(I,1)=ITEMGRP(N,I+1)+TOHI(I,1). 

-ELSE. 

-END IF. 

-END LOOP. 

END LOOP. 

 

LOOP I=1 TO NITEM. 

-LOOP N=1 TO NCASE. 

-DO IF(ITEMGRP(N,1)=3). 

-        COMPUTE TOLO(I,1)=ITEMGRP(N,I+1)+TOLO(I,1). 

-ELSE. 

-END IF. 

-END LOOP. 

END LOOP. 

 

COMPUTE DINDEX=TOHI/#UP-TOLO/#LOW. 

COMPUTE PVALUET=CSUM(SCORE)/NCASE. 

COMPUTE PVALUE=TRANSPOS(PVALUET). 

COMPUTE PD={PVALUE,DINDEX}. 

**********************************. 

*Change file location 'C:\temp'. 

**********************************. 

 

SAVE PD 

 /OUTFILE='C:\temp\PD.SAV' 

 /VARIABLES=PVALUE D_INDEX. 

 

END MATRIX. 

******************************************************************************************** 

**********************************. 

*Change file location 'C:\temp'. 

**********************************. 

GET FILE='C:\temp\TOTSCORE.SAV'. 

TITLE '****************************************************'. 

TITLE '***TEST STATISTICS--TOTAL SCORE DISTRIBUTION***'. 

TITLE '****************************************************'. 

EXECUTE. 

VARIABLE LABELS 

TOTSC 'TOTAL SCORE'. 

DESCRIPTIVES 

  VARIABLES=TOTSC 

  /STATISTICS=MEAN STDDEV MIN MAX SKEWNESS KURTOSIS. 

EXECUTE. 

 

**********************************. 

*Change file location 'C:\temp'. 

**********************************. 

GET FILE='C:\temp\ITEMSCORE.SAV'. 

TITLE '****************************************************'. 

TITLE '************RELIABILITY ANALYSIS*****************'. 

TITLE '****************************************************'. 

EXECUTE. 

RELIABILITY 

  /VARIABLES=ALL 

  /FORMAT=NOLABELS 

  /SCALE(ALPHA)=ALL/MODEL=ALPHA 

  /SUMMARY=TOTAL CORR. 

EXECUTE. 

**********************************. 

*Change file location 'C:\temp'. 

**********************************. 

GET FILE='C:\temp\PD.SAV'. 

TITLE '********************************************************'. 

TITLE '********P VALUE AND DISCRIMINATION INDEX***********'. 

TITLE '********************************************************'. 

EXECUTE. 

COMPUTE ITEM_NUM=$CASENUM. 

FORMAT ITEM_NUM (F5.0). 

 

LIST VARIABLES=ITEM_NUM PVALUE D_INDEX. 

EXECUTE. 

**********************************. 

*Change file location 'C:\temp'. 

**********************************. 

GET FILE='C:\temp\RAWGRP.SAV'. 

TITLE '********************************************************'. 

TITLE '************DISTRACTOR ANALYSIS*********************'. 

TITLE '********************************************************'. 

EXECUTE. 

 

VARIABLE LABELS 

GRP 'GROUP'. 

 

VALUE LABELS 

GRP 1 'HIGH' 3 'LOW'. 

 

USE ALL. 

COMPUTE filter_$=(GRP>0). 

VARIABLE LABEL filter_$ 'GRP>0 (FILTER)'. 

VALUE LABELS filter_$  0 'Not Selected' 1 'Selected'. 

FORMAT filter_$ (f1.0). 

FILTER BY filter_$. 

EXECUTE . 

FORMAT ALL (F1.0). 

MISSING VALUE ALL (-9). 

 

****************************************************************. 

*Change variable names:list all your item names after "BY" 

*The abbreviated form does not work here. 

*****************************************************************. 

CROSSTABS 

  /TABLES=GRP BY MC1 MC2 MC3 MC4 MC5 MC6 MC7 MC8 MC9 MC10 MC11 MC12 MC13 MC14 MC15  

  /FORMAT= AVALUE TABLES 

  /CELLS= COUNT 

  /COUNT ROUND CELL. 

Appendix B 
CTTITEM_P.sps -- SPSS syntax file for polytomous items 

DATA LIST FILE='C:\temp\polyitem.DAT' 

   / IT1 TO IT22 1-22. 

MISSING VALUES ALL ( ). 

 

SET MXLOOPS=300 /MPRINT=NO. 

 

MATRIX. 

GET RAW 

  /VARIABLES=IT1 TO IT22  

  /FILE=*  

  /MISSING=ACCEPT 

  /SYSMIS=-9. 

 

COMPUTE NCASE=NROW(RAW)-2. 

COMPUTE NITEM=NCOL(RAW). 

COMPUTE SCORE=MAKE(NCASE,NITEM,0). 

 

LOOP N=3 TO NROW(RAW). 

- LOOP I=1 TO NITEM. 

- DO IF RAW(N,I)=-9. 

- COMPUTE SCORE(N-2,I)=-9. 

- ELSE IF RAW(2,I)=1. 

-  COMPUTE SCORE((N-2),I)=RAW(N,I). 

- ELSE IF RAW(2,I)=0. 

-  COMPUTE SCORE((N-2),I)=RAW(1,I)+1-RAW(N,I). 

-END IF. 

-END LOOP. 

END LOOP. 

 

******COMPUTE TOTAL SCORE 

 

SAVE SCORE 

/OUTFILE='C:\temp\ITEMSCORE.SAV' 

/VARIABLES=IT1 TO IT22. 

 

END MATRIX. 

******************************************************************************************** 

******************************************************************************************** 

GET FILE='C:\temp\ITEMSCORE.SAV'. 

TITLE '****************************************************'. 

TITLE '******TOTAL SCORE DISTRIBUTION******************'. 

TITLE '****************************************************'. 

EXECUTE. 

 

COMPUTE TOTSC=SUM(IT1 TO IT22).  

 

DESCRIPTIVES 

  VARIABLES=TOTSC 

  /STATISTICS=MEAN STDDEV MIN MAX SKEWNESS KURTOSIS. 

EXECUTE. 

TITLE '****************************************************'. 

TITLE '************RELIABILITY ANALYSIS*****************'. 

TITLE '****************************************************'. 

EXECUTE. 

 

MISSING VALUE ALL (-9). 

 

RELIABILITY 

  /VARIABLES= IT1 TO IT22 

  /FORMAT=NOLABELS 

  /SCALE(ALPHA)=ALL/MODEL=ALPHA 

  /SUMMARY=TOTAL CORR . 

EXECUTE. 

TITLE '*******************************************************'. 

TITLE '***********ITEM MEAN FOR EVERY ITEM****************'. 

TITLE '*******************************************************'. 

EXECUTE. 

DESCRIPTIVES 

  VARIABLES= IT1 TO IT22 

  /STATISTICS=MEAN. 

EXECUTE. 

FORMAT ALL (F1.0). 

TITLE '*********************************************************'. 

TITLE '***********SCORE FREQENCY DISTRIBUTION**************'. 

TITLE '*********************************************************'. 

EXECUTE. 

FREQUENCIES 

  VARIABLES= IT1 TO IT22. 

EXECUTE. 

 

 


